Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematical Sciences
goutzicj@oneonta.edu

Lesson 7

Areas between curves

Initializations

```
> restart;
  with(plots):
>
```

7.1 Areas between curves

Examples

Example 7.1.1

Find the area of the region enclosed by the curves

$$y = x^2 - 4 \quad \text{and} \quad y = 3x + 5$$

Solution

First, we compute the points of intersection of the curves and make a sketch. Computing the points of intersection before plotting the graph has the advantage that we now have a reasonable idea about the domain necessary to obtain a good visualization of the problem.

```
> f:=x->x^2+4;
  g:=x->3*x+5;
  poi:=[solve(f(x)=g(x), x)];
  evalf(poi);
```

$$f := x \rightarrow x^2 + 4$$

$$g := x \rightarrow 3\,x + 5$$

$$poi := \left[\frac{3}{2} + \frac{1}{2}\sqrt{13}, \frac{3}{2} - \frac{1}{2}\sqrt{13} \right]$$

$$[3.302775638, -0.302775638] \qquad \textbf{(2.1.1.1)}$$

```
> plot([f(x), g(x)], x=-1..4, color=[red, blue]);
```



Observe that color coordination allows for easy association of curves and formulas. The desired area is readily computed.

```
> e1:=Int(g(x)-f(x), x=poi[2]..poi[1]);
```

$$e1 := \int_{\frac{3}{2}-\frac{1}{2}\sqrt{13}}^{\frac{3}{2}+\frac{1}{2}\sqrt{13}} \left(-x^2+1+3x\right)\,dx \qquad \textbf{(2.1.1.2)}$$

```
> area:=simplify(value(e1));
  evalf(area);
```

$$area := \frac{13}{6}\sqrt{13}$$

$$7.812027764 \qquad \textbf{(2.1.1.3)}$$

```
>
```

Example 7.1.2

Find a decimal approximation for the area enclosed by the curves

$$y = x^4 - 1 \quad \textbf{and} \quad y = x \sin x^2$$

Solution

Since it is not mathematically possible to compute a closed form expression for the points of intersection of these curves, we have no choice but first to make a sketch.

```
> f:=x->x^4-1;
  g:=x->x*sin(x^2);
  plot([f(x), g(x)], x=-2..2, y=-2..2, color=[red, blue]);
```

$$f := x \rightarrow x^4 - 1$$

$$g := x \rightarrow x \sin\left(x^2\right)$$



Information from the graph about the approximate location of the points of intersection can be supplied to Maple's numeric equation solver **fsolve**.

```
> xpoi[1]:=fsolve(f(x)=g(x), x=-1..0);
  xpoi[2]:=fsolve(f(x)=g(x), x=1..2);
```

$$xpoi_1 := -0.8294465625$$

$$xpoi_2 := 1.220087400 \qquad \textbf{(2.1.2.1)}$$

With this information the area of the enclosed region can easily be computed.

```
> e1:=Int(g(x)-f(x), x=xpoi[1]..xpoi[2]);
```

$$e1 := \int_{-0.8294465625}^{1.220087400} \left(x \sin\left(x^2\right) - x^4 + 1\right) \mathrm{d}x \qquad \textbf{(2.1.2.2)}$$

```
> area:=value(e1);
```

$$area := 1.775499495 \qquad \textbf{(2.1.2.3)}$$

```
>
```

Example 7.1.3

Compute the area of the region enclosed by the two parabolas

$$x = y^2 - 5y - 4 \quad \textbf{and} \quad x = -2\,y^2 + 7y - 1$$

First use an integration over *x*, then use an integration over *y*.

Solution

The computation will show that this is a kind of problem in which integration over *y* is highly preferable. Code the curves, locate the points of intersection, and make a sketch.

```
> c1:=x=y^2-5*y-4;
  c2:=x=-2*y^2+7*y-1;
```

$$c1 := x = y^2 - 5\,y - 4$$

$$c2 := x = -2\,y^2 + 7\,y - 1 \qquad \textbf{(2.1.3.1)}$$

```
> e1:=solve({c1, c2}, {x, y});
```

$$e1 := \{y = \mathrm{RootOf}(\_Z^2 - 4\,\_Z - 1, label = \_L1), x = -3 - \mathrm{RootOf}(\_Z^2 - 4\,\_Z - 1, label = \_L1)\}$$ **(2.1.3.2)**

The **RootOf** placeholders can be evaluated using the allvalues command.

```
> poi:=allvalues(e1);
```

$$poi := \{y = 2 + \sqrt{5}, x = -5 - \sqrt{5}\}, \{y = 2 - \sqrt{5}, x = -5 + \sqrt{5}\}$$ **(2.1.3.3)**

```
> evalf(poi);
```

$$\{y = 4.236067977, x = -7.236067977\}, \{y = -0.236067977, x = -2.763932023\}$$ **(2.1.3.4)**

```
> implicitplot([c1, c2], x=-12..8, y=-1..5, color=[red,
  blue]);
```



Observe that **c1** is the curve which opens to the right. We now compute the smallest and the largest *x*-coordinate in the region of which we want to find the area.

```
> yextr[1]:=solve(diff(rhs(c1), y)=0, y);
  xmin:=subs(y=yextr[1], rhs(c1));
```

$$yextr_1 := \frac{5}{2}$$

$$xmin := -\frac{41}{4}$$ **(2.1.3.5)**

```
> yextr[2]:=solve(diff(rhs(c2), y)=0, y);
  xmax:=subs(y=yextr[2], rhs(c2));
```

$$yextr_2 := \frac{7}{4}$$

$$xmax := \frac{41}{8} \quad \textbf{(2.1.3.6)}$$

Code the *x*-coordinates of the points of intersection and generate the top and bottom halves of the curves.

```
> X[1]:=subs(poi[1], x);
  X[2]:=subs(poi[2], x);
```

$$X_1 := -5 - \sqrt{5}$$

$$X_2 := -5 + \sqrt{5} \quad \textbf{(2.1.3.7)}$$

```
> TBc1:=solve(c1, y);
  TBc2:=solve(c2, y);
```

$$TBc1 := \frac{5}{2} + \frac{1}{2}\sqrt{41 + 4x}, \frac{5}{2} - \frac{1}{2}\sqrt{41 + 4x}$$

$$TBc2 := \frac{7}{4} + \frac{1}{4}\sqrt{41 - 8x}, \frac{7}{4} - \frac{1}{4}\sqrt{41 - 8x} \quad \textbf{(2.1.3.8)}$$

Express the area of the enclosed region as a sum of three integrals.

```
> S1:=Int(TBc1[1]-TBc1[2], x=xmin..X[1])+Int(TBc2[1]-TBc1
  [2], x=X[1]..X[2])+Int(TBc2[1]-TBc2[2], x=X[2]..xmax);
```

$$S1 := \int_{-\frac{41}{4}}^{-5-\sqrt{5}} \sqrt{41 + 4x}\, dx + \int_{-5-\sqrt{5}}^{-5+\sqrt{5}} \left(-\frac{3}{4} + \frac{1}{4}\sqrt{41 - 8x} + \frac{1}{2}\sqrt{41 + 4x}\right) dx + \int_{-5+\sqrt{5}}^{\frac{41}{8}} \frac{1}{2}\sqrt{41 - 8x}\, dx \quad \textbf{(2.1.3.9)}$$

Maple makes quick work of the evaluation of this most unpleasant formula.

```
> Area[`by integration over x`]:=simplify(value(S1));
```

$$Area_{by\ integration\ over\ x} := 20\sqrt{5} \quad \textbf{(2.1.3.10)}$$

The same result can be obtained in merely a few lines, by integrating over y.

```
> ymin:=subs(poi[2], y);
  ymax:=subs(poi[1], y);
```

$$ymin := 2 - \sqrt{5}$$

$$ymax := 2 + \sqrt{5} \quad \textbf{(2.1.3.11)}$$

```
> S2:=Int(rhs(c2)-rhs(c1), y=ymin..ymax);
```

$$S2 := \int_{2-\sqrt{5}}^{2+\sqrt{5}} \left(-3y^2 + 12y + 3\right) dy \quad \textbf{(2.1.3.12)}$$

```
> Area[`by integration over y`]:=simplify(value(S2));
```

$$Area_{by\ integration\ over\ y} := 20\sqrt{5} \quad (2.1.3.13)$$

```
>
```